

BALLS AND URNS, ETC.

Philippe Flajolet

(Version of December 14, 1996)

Balls and urns models are basic in combinatorics, statistics, analysis of algorithms, and statistical physics. These models are nicely decomposable and their basic properties can be explored using tools developed for the automatic manipulation of combinatorial models, like [Combstruct](#). As is well-known there are four types of models, depending on whether balls and urns are taken to be distinguishable or not.

■ The four basic models

We consider the placement of balls into urns in all possible ways. For definiteness, we examine only the situation of nonempty urns, so that the number of possible configurations of a fixed size (i.e., a fixed number of balls) is always finite. If the balls are distinguishable, we may assume them to be numbered consecutively by integers $1, 2, \dots, n$; in this case, we are dealing with labelled structures, and balls are labelled atoms. If the balls are indistinguishable, then we simply regard them as anonymous unlabelled atoms (generically called Z , by a global convention of Combstruct). If the urns are distinguishable, we may view them as arranged in a row, so that we are dealing with a [Sequence](#) construction; otherwise, we have a [Set](#) construction. (The [Set](#) construction of Combstruct means a multiset, that is to say a set where repetitions are allowed.) Balls are not ordered within an urn, so that an urn is a priori a [Set](#) of balls. This gives rise to four different models:

- DBDU: distinguishable balls and distinguishable urns; we are dealing with Sequences of Sets, in a labelled universe;
- DBIU: distinguishable balls and indistinguishable urns; we are dealing with Sets of Sets, in a labelled universe;
- IBDU: indistinguishable balls and distinguishable urns; we are dealing with Sequences of Sets, in an unlabelled universe;
- IBIU: indistinguishable balls and indistinguishable urns; we are dealing with Sets of Sets, in an unlabelled universe.

In combstruct, this is expressed by four different, but similar looking, specifications:

```
> with(combstruct);  
[allstructs, count, draw, finished, gfeqns, gfseries, gfsolve, iterstructs, nextstruct,  
  prog_gfeqns, prog_gfseries, prog_gfsolve]  
[> DBDU := [S, {S=Sequence(U), U=Set(Z, card>=1)}, labelled] :  
[> DBIU := [S, {S=Set(U), U=Set(Z, card>=1)}, labelled] :  
[> IBDU := [S, {S=Sequence(U), U=Set(Z, card>=1)}, unlabelled] :  
[> IBIU := [S, {S=Set(U), U=Set(Z, card>=1)}, unlabelled] :  
[> for spec in DBDU, DBIU, IBDU, IBIU do draw(spec, size=10) od;  
  Sequence(Set(Z4), Set(Z3), Set(Z5), Set(Z9, Z7), Set(Z2), Set(Z8), Set(Z1), Set(Z10, Z6))
```

$\text{Set}(\text{Set}(Z_5), \text{Set}(Z_2, Z_8, Z_{10}), \text{Set}(Z_3, Z_9, Z_7), \text{Set}(Z_4, Z_1, Z_6))$

$\text{Sequence}(\text{Set}(Z, Z), \text{Set}(Z, Z, Z, Z, Z, Z, Z), \text{Set}(Z))$

$\text{Set}(\text{Set}(Z), \text{Set}(Z), \text{Set}(Z, Z), \text{Set}(Z, Z), \text{Set}(Z, Z, Z, Z))$

The corresponding counting sequences satisfy natural domination conditions that one can summarize by the informal inequality: "Distinguishable>Indistinguishable"

```
> for spec in DBDU, DBIU, IBDU, IBIU do
  seq(count(spec, size=j), j=1..12) od;
1, 3, 13, 75, 541, 4683, 47293, 545835, 7087261, 102247563, 1622632573, 28091567595
1, 2, 5, 15, 52, 203, 877, 4140, 21147, 115975, 678570, 4213597
1, 2, 4, 8, 16, 32, 64, 128, 256, 512, 1024, 2048
1, 2, 3, 5, 7, 11, 15, 22, 30, 42, 56, 77
```

In the sequel, it is convenient to represent objects by a more concise notation. We thus introduce "reduction" procedures for labelled and unlabelled objects:

```
> lreduce:=proc(e) eval(subs({Set=proc() {args} end,
  Sequence=proc() [args] end}, e)) end:
  urreduce:=proc(e) eval(subs({Set=proc() {[args]} end,
  Sequence=proc() [args] end}, e)) end:
```

Since the set construction "{}" in Maple does not keep multisets, an unlabelled (multi)set will be represented as "{[...]}".

```
> for spec in DBDU, DBIU do lreduce(draw(spec, size=25)) od;
  for spec in IBDU, IBIU do urreduce(draw(spec, size=25)) od;
[{{Z24}, {Z11}, {Z14}, {Z5}, {Z25}, {Z7, Z20}, {Z2}, {Z4}, {Z13}, {Z17}, {Z1}, {Z3},
  {Z18}, {Z10}, {Z23}, {Z9}, {Z21}, {Z19}, {Z22}, {Z8, Z15}, {Z12}, {Z6}, {Z16}}
[{{Z2}, {Z19}, {Z9, Z21}, {Z1, Z11}, {Z4, Z20}, {Z24, Z12}, {Z3, Z10, Z22, Z15}, {Z13, Z23},
  {Z8, Z17, Z18}, {Z5, Z7, Z14, Z16}, {Z6, Z25}}
[{{[Z, Z, Z, Z, Z, Z]}, {[Z]}, {[Z]}, {[Z]}, {[Z, Z, Z]}, {[Z, Z, Z]}, {[Z, Z]}, {[Z]},
  {[Z]}, {[Z, Z, Z]}, {[Z, Z, Z]}}
[{{[Z, Z, Z]}, {[Z, Z, Z]}, {[Z, Z, Z, Z, Z, Z, Z, Z, Z, Z, Z, Z, Z, Z, Z, Z, Z, Z]}, {[Z, Z]}}]
```

On such simulations, we see that there tends to be fewer urns in models of type IU, but more filled ones.

■ Distinguishable balls (labelled structures)

■ Distinguishable urns

In this model, we deal with distinguishable balls (labelled atoms) that go in all possible way into distinguishable urns corresponding to the specification:

```
[ > DBDU := [S, {S=Sequence(U), U=Set(Z, card>=1)}, labelled] :
```

Combinatorially, this model is the same as of Surjections from $[1..n]$ to an initial segment of the integers. It is the one that leads to larger cardinality counts.

```
> for j to 3 do j:=map(lreduce, allstructs(DBDU, size=j)) od;
1 = [{{Z1}}]
2 = [{{Z2}, {Z1}}, {{Z1}, {Z2}}, {{Z2, Z1}}]
```

```
3 = [[{Z3, Z2, Z1}], [{Z1}, {Z3}, {Z2}], [{Z2}, {Z1}, {Z3}], [{Z3, Z2}, {Z1}],
    [{Z3}, {Z2, Z1}], [{Z3, Z1}, {Z2}], [{Z1}, {Z3, Z2}], [{Z2}, {Z3, Z1}],
    [{Z3}, {Z1}, {Z2}], [{Z3}, {Z2}, {Z1}], [{Z2}, {Z3}, {Z1}], [{Z2, Z1}, {Z3}],
    [{Z1}, {Z2}, {Z3}]]
```

```
> seq(count(DBDU, size=j), j=0..30);
```

```
1, 1, 3, 13, 75, 541, 4683, 47293, 545835, 7087261, 102247563, 1622632573,
28091567595, 526858348381, 10641342970443, 230283190977853,
5315654681981355, 130370767029135901, 3385534663256845323,
92801587319328411133, 2677687796244384203115, 81124824998504073881821,
2574844419803190384544203, 85438451336745709294580413,
2958279121074145472650648875, 106697365438475775825583498141,
4002225759844168492486127539083, 155897763918621623249276226253693,
6297562064950066033518373935334635,
263478385263023690020893329044576861,
11403568794011880483742464196184901963
```

Such tables are quite useful for checking various combinatorial conjectures. Here, we may verify that these numbers are the sequence **M2952** of the *Encyclopedia of Integer Sequences* by Sloane and Plouffe, where they are known as the numbers of preferential arrangements of n things.

The counting problem is solved automatically by [combstruct\[geqns\]](#), [combstruct\[gfseries\]](#) (a series alternative to [combstruct\[count\]](#)) and [combstruct\[gsolve\]](#):

```
> gfeqns(op(2..3, DBDU), z);
```

$$\left[U(z) = e^{Z(z)} - 1, S(z) = \frac{1}{1 - U(z)}, Z(z) = z \right]$$

```
> Order:=12: gfseries(op(2..3, DBDU), z);
```

```
table([
```

$$S(z) = 1 + z + \frac{3}{2}z^2 + \frac{13}{6}z^3 + \frac{25}{8}z^4 + \frac{541}{120}z^5 + \frac{1561}{240}z^6 + \frac{47293}{5040}z^7 + \frac{36389}{2688}z^8 + \frac{7087261}{362880}$$

$$z^9 + \frac{34082521}{1209600}z^{10} + \frac{1622632573}{39916800}z^{11} + O(z^{12})$$

$$U(z) = z + \frac{1}{2}z^2 + \frac{1}{6}z^3 + \frac{1}{24}z^4 + \frac{1}{120}z^5 + \frac{1}{720}z^6 + \frac{1}{5040}z^7 + \frac{1}{40320}z^8 + \frac{1}{362880}z^9 +$$

$$\frac{1}{3628800}z^{10} + \frac{1}{39916800}z^{11} + O(z^{12})$$

$$Z(z) = z$$

```
])
```

```
> gfsolve(op(2..3, DBDU), z);
```



```

    {{Z1}, {Z2}, {Z3}}]
4 = [{{Z3, Z2, Z1}, {Z4}}, {{Z2}, {Z3}, {Z1, Z4}}, {{Z1}, {Z2}, {Z3, Z4}},
    {{Z2}, {Z3, Z1, Z4}}, {{Z1}, {Z3, Z2, Z4}}, {{Z1}, {Z2}, {Z3}, {Z4}},
    {{Z2, Z1}, {Z3}, {Z4}}, {{Z1}, {Z3}, {Z2, Z4}}, {{Z3, Z2, Z1, Z4}},
    {{Z2, Z1}, {Z3, Z4}}, {{Z3}, {Z2, Z1, Z4}}, {{Z3, Z1}, {Z2, Z4}},
    {{Z1}, {Z3, Z2}, {Z4}}, {{Z3, Z2}, {Z1, Z4}}, {{Z2}, {Z3, Z1}, {Z4}}]
> seq(count(DBIU, size=j), j=0..30);
1, 1, 2, 5, 15, 52, 203, 877, 4140, 21147, 115975, 678570, 4213597, 27644437,
190899322, 1382958545, 10480142147, 82864869804, 682076806159,
5832742205057, 51724158235372, 474869816156751, 4506715738447323,
44152005855084346, 445958869294805289, 4638590332229999353,
49631246523618756274, 545717047936059989389, 6160539404599934652455,
71339801938860275191172, 846749014511809332450147

```

Such tables are quite useful for checking various combinatorial conjectures. Here, we may verify that these numbers are the sequence **M1484** of the *Encyclopedia of Integer Sequences* by Sloane and Plouffe. They are the well-known [Bell numbers](#) of combinatorial theory that also appear as moments of the Poisson distribution, in the calculus of finite differences, etc.

We automatically obtain the exponential generating function as

```

> gfsolve(op(2..3, DBIU), z);
      {Z(z) = z, U(z) = e^z - 1, S(z) = e^(e^z - 1)}
> P_z := subs("", S(z));
      P_z := e^(e^z - 1)
> series(P_z, z=0, 8);
      1 + z + z^2 + 5/6 z^3 + 5/8 z^4 + 13/30 z^5 + 203/720 z^6 + 877/5040 z^7 + O(z^8)
> Order:=8: gfsseries(op(2..3, DBIU), z);
table([
      S(z) = 1 + z + z^2 + 5/6 z^3 + 5/8 z^4 + 13/30 z^5 + 203/720 z^6 + 877/5040 z^7 + O(z^8)
      U(z) = z + 1/2 z^2 + 1/6 z^3 + 1/24 z^4 + 1/120 z^5 + 1/720 z^6 + 1/5040 z^7 + O(z^8)
      Z(z) = z
])

```

By expanding and truncating, we obtain excellent approximations (this is in fact a version of a formula found by Dobinski in 1877):

```

> P_n_asympt := exp(-1) * Sum(k^n/k!, k=0..2*n);
for j by 3 to 20 do
j, evalf(count(DBIU, size=j)/subs(n=j, P_n_asympt), 30); od;

```

$$P_{n_asympt} := e^{(-1)} \left(\sum_{k=0}^{2n} \frac{k^n}{k!} \right)$$

1, 1.35914091422952261768014373568
 4, 1.00052146464862250684406987918
 7, 1.00000006071729267635177431357
 10, 1.00000000000111597855712643959
 13, 1.00000000000000000525280524286
 16, 1.000000000000000000000839324
 19, 1.00000000000000000000000000001

Indistinguishable balls (unlabelled structures)

Distinguishable urns (integer compositions)

We start from the specification

```
[ > IBDU := [S, {S=Sequence(U), U=Sequence(Z, card>=1)}, unlabelled]
      :
```

In this particular case, as balls are indistinguishable, we may as well consider urns as [Sequence](#) of atoms. The reason for doing this is a simpler form of generating functions (as we do not have to go unnecessarily through Polya operators) as well as faster computations.

We can check that this new version is equivalent to the earlier one, namely

```
[ > IBDU_0 := [S, {S=Sequence(U), U=Set(Z, card>=1)}, unlabelled] :
  > seq(count(IBDU, size=j), j=0..20);
    seq(count(IBDU_0, size=j), j=0..20);
  1, 1, 2, 4, 8, 16, 32, 64, 128, 256, 512, 1024, 2048, 4096, 8192, 16384, 32768, 65536,
    131072, 262144, 524288
  1, 1, 2, 4, 8, 16, 32, 64, 128, 256, 512, 1024, 2048, 4096, 8192, 16384, 32768, 65536,
    131072, 262144, 524288
```

Of course, here we recognize the powers of two: the result is combinatorially obvious since a partition can be obtained by inserting arbitrary cuts in the integer interval 1, ..., n. We can also check this with `combstruct[gsolve]`

```
[ > gfsolve(op(2..3, IBDU), z);
      { Z(z) = z, S(z) = \frac{-1+z}{-1+2z}, U(z) = -\frac{z}{-1+z} }
  > SS_z := subs(" ", S(z));
      SS_z := \frac{-1+z}{-1+2z}
  > series(SS_z, z=0, 10);
      1 + z + 2 z^2 + 4 z^3 + 8 z^4 + 16 z^5 + 32 z^6 + 64 z^7 + 128 z^8 + 256 z^9 + O(z^10)
```

Indistinguishable urns (integer partitions)

We start from the specification

```
[ > IBIU := [S, {S=Set(U), U=Sequence(Z, card>=1)}, unlabelled] :
```

Combinatorially, we are specifying integer partitions that describe the occupancy profile of urns.

```

> for j to 6 do j:=map(ureduce,allstructs(IBIU,size=j)) od;
      1 = {[[[Z]]]}
      2 = {[[[Z], [Z]], [[Z, Z]]}
      3 = {[[[Z, Z, Z]], [[Z], [Z], [Z]], [[Z, Z], [Z]]}
      4 = {[[[Z, Z], [Z, Z]], [[Z, Z, Z], [Z]], [[Z, Z], [Z], [Z]],
          [[Z], [Z], [Z], [Z]], [[Z, Z, Z, Z]]}
      5 = {[[[Z, Z, Z], [Z, Z]], [[Z, Z], [Z, Z], [Z]], [[Z, Z], [Z], [Z], [Z]],
          [[Z, Z, Z, Z, Z]], [[Z], [Z, Z, Z, Z]], [[Z, Z, Z], [Z], [Z]],
          [[Z], [Z], [Z], [Z], [Z]]}
      6 = {[[[Z], [Z], [Z], [Z], [Z], [Z]], [[Z, Z, Z], [Z, Z, Z]],
          [[Z, Z, Z], [Z], [Z], [Z]], [[Z, Z], [Z, Z], [Z], [Z]], [[Z, Z], [Z, Z], [Z, Z]],
          [[Z], [Z], [Z, Z, Z, Z]], [[Z, Z], [Z], [Z], [Z], [Z]], [[Z, Z, Z], [Z, Z], [Z]],
          [[Z], [Z, Z, Z, Z, Z]], [[Z, Z, Z, Z, Z, Z]], [[Z, Z], [Z, Z, Z, Z]]}

```

Naturally, since we are dealing with sets of summands (the order does not count), we may as well regard these objects as an increasing sequence of summands that sum to the size n , or equivalently as "staircases" with size being the area below the staircase.

```

> preduce:=proc(e) sort(eval(subs({Set=proc() [args] end,
Sequence=proc() nargs end},e))) end:
> rand_part:=draw(IBIU,size=100): ureduce(rand_part);
preduce(rand_part);
{[[[Z, Z], [Z, Z], [Z, Z], [Z, Z], [Z, Z], [Z, Z], [Z], [Z], [Z], [Z], [Z], [Z], [Z], [Z],
[Z], [Z], [Z], [Z], [Z], [Z], [Z], [Z], [Z], [Z, Z, Z, Z, Z], [Z, Z, Z, Z, Z],
[Z, Z, Z, Z, Z], [Z, Z, Z, Z, Z], [Z, Z, Z, Z, Z], [Z, Z, Z, Z, Z], [Z, Z, Z, Z, Z],
[Z, Z, Z, Z, Z], [Z, Z, Z, Z, Z, Z, Z, Z, Z, Z, Z, Z, Z, Z, Z, Z, Z, Z],
[Z, Z, Z, Z, Z, Z, Z, Z, Z, Z, Z, Z, Z, Z]]}
[1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 2, 2, 2, 2, 2, 2, 5, 5, 5, 5, 5, 5, 5, 5, 14, 17]

```

There are much fewer structures than in previous models:

```

> seq(count( IBIU, size=j ), j=0..30);
1, 1, 2, 3, 5, 7, 11, 15, 22, 30, 42, 56, 77, 101, 135, 176, 231, 297, 385, 490, 627, 792,
1002, 1255, 1575, 1958, 2436, 3010, 3718, 4565, 5604

```

The random generation process is nontrivial as one must generate objects up to certain symmetries. The first time, counting tables are set up on the fly, so that random generation takes a few seconds for size $n \leq 100$.

```

> for i from 0 by 20 to 100 do i,preduce(draw( IBIU, size=i ));
od;
      0, E
      20, [1, 1, 1, 1, 1, 3, 4, 8]
      40, [1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 2, 2, 2, 2, 2, 2, 2, 2, 4, 4, 6]

```

60, [1, 1, 2, 2, 2, 2, 2, 3, 4, 5, 5, 6, 12, 13]
 80, [1, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 3, 3, 3, 3, 3, 3, 3, 3, 4, 4, 5, 7, 13]
 100, [1, 1, 1, 1, 1, 3, 3, 3, 3, 3, 3, 5, 5, 5, 10, 10, 11, 31]

Next, random generation becomes faster:

```
> for i to 10 do preduce(draw(IBIU,size=100)); od;
[1, 1, 1, 1, 1, 1, 1, 1, 2, 6, 10, 12, 16, 22, 24]
[1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 2, 2, 2, 2, 2, 2, 3, 3, 3, 4, 4, 5, 9, 9, 17, 20]
[1, 1, 1, 1, 1, 1, 2, 3, 4, 4, 9, 11, 11, 18, 32]
[1, 1, 1, 1, 1, 2, 2, 2, 2, 2, 4, 7, 7, 10, 12, 20, 25]
[1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 2, 2, 2, 3, 3, 3, 3, 3, 3, 4, 4, 4, 4, 4, 4, 4, 4, 10, 19]
[4, 4, 4, 4, 5, 6, 6, 9, 18, 20, 20]
[1, 1, 1, 2, 2, 2, 2, 2, 2, 2, 3, 5, 5, 5, 5, 6, 6, 6, 6, 7, 14, 15]
[1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 2, 2, 2, 2, 3, 4, 5, 9, 10, 12, 13, 22]
[1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 2, 2, 2, 3, 3, 4, 5, 5, 5, 7, 8, 12, 12, 18]
[1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 4, 4, 7, 14, 23, 28]
```

In this particular case, the random generation procedure that is automatically built by Combstruct coincides with a method especially designed by Wilf for integer partitions. By design, Combstruct accepts in full generality arbitrary compositions of Set and Cycle constructions (in addition to Union, Product, Sequence, etc).

The generating functions are now more complicated since they involve Polya operators.

```
> gfeqns(op(2..3,IBIU),z);
[
  [
    U(z) = 1 / (1 - Z(z)) - 1, S(z) = e^{sum_{j_1=1}^{\infty} U(z)^{j_1} / j_1}, Z(z) = z
  ]
]
> gfsolve(op(2..3,IBIU),z);
[
  [
    Z(z) = z, U(z) = -z / (-1 + z), S(z) = e^{sum_{j_1=1}^{\infty} (-z / (-1 + z))^{j_1} / j_1}
  ]
]
```

Such Polya operators are however known to combstruct[gfseries]

```
> gfseries(op(2..3,IBIU),z);
table([
  S(z) = 1 + z + 2 z^2 + 3 z^3 + 5 z^4 + 7 z^5 + 11 z^6 + 15 z^7 + O(z^8)
  U(z) = z + z^2 + z^3 + z^4 + z^5 + z^6 + z^7 + O(z^8)
  Z(z) = z
])
```

though they are not otherwise "known" to Maple

```
> subs("",S(z)); series("",z=0);
```


$$e^{\left(\sum_{j_1=1}^{\infty} \left(-\frac{z^{j_1}}{\binom{j_1-1}{j_1}} \right) \right)}$$

Error, (in series/exp) unable to compute series

In such cases, one has to resort either to simplification by hand (not always possible) or to the literature. Here, it is very well known that the generating function of integer partitions is

```
> PP_z:=Product(1/(1-z^k),k=1..infinity);
```

$$PP_z := \prod_{k=1}^{\infty} \frac{1}{1-z^k}$$

```
> Order:=12: series(subs(infinity=Order+2,PP_z),z=0);
```

```
1+z+2z^2+3z^3+5z^4+7z^5+11z^6+15z^7+22z^8+30z^9+42z^10+56z^11+O(z^12)
```

Constrained models

Number of urns in surjections

The approach developed so far may be tuned to analyse a variety parameters. We explore here the way Combstruct may serve to analyse the number of urns as well as related situations with bounded urn capacity. We focus on counts and building numerical tables. Naturally, random generation and exhaustive listing are possible from any of these specifications.

We deal here with a fixed number of urns in the model DBDU that corresponds to surjections. Combinatorial specifications may actually be computed in Maple, then used by Combstruct.

The following procedure computes the specifications with r urns.

```
> surj := [S, {S=Sequence(U, card=r), U=Set(Z, card>=1)},
  labelled] : subs(r=5, surj);
```

```
[S, {U=Set(Z, 1 ≤ card), S=Sequence(U, card=5)}, labelled]
```

In passing, this illustrates the use of cardinality modifiers for Sequence, Set, and Cycle constructions.

The following counts imply the first few values of the probability distribution of the number of urns in a random unconstrained surjection:

```
> for i to 5 do seq(count(subs(r=i, surj), size=m), m=0..10)
  od;
```

```
0, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1
```

```
0, 0, 2, 6, 14, 30, 62, 126, 254, 510, 1022
```

```
0, 0, 0, 6, 36, 150, 540, 1806, 5796, 18150, 55980
```

```
0, 0, 0, 0, 24, 240, 1560, 8400, 40824, 186480, 818520
```

```
0, 0, 0, 0, 0, 120, 1800, 16800, 126000, 834120, 5103000
```

and we may build tables of probability distributions automatically:

```
> for i to 7 do
  seq(evalf(count(subs(r=i, surj), size=m)/count(DBDU, size=m),
  4), m=0..10) od;
```

0, 1., .3333, .07692, .01333, .001848, .0002135, .00002114, .1832 10⁻⁵, .1411 10⁻⁶,
 .9780 10⁻⁸

0, 0, .6667, .4615, .1867, .05545, .01324, .002664, .0004653, .00007196, .9995 10⁻⁵

0, 0, 0, .4615, .4800, .2773, .1153, .03819, .01062, .002561, .0005475

0, 0, 0, 0, .3200, .4436, .3331, .1776, .07479, .02631, .008005

0, 0, 0, 0, 0, .2218, .3844, .3552, .2308, .1177, .04991

0, 0, 0, 0, 0, 0, .1537, .3197, .3509, .2688, .1607

0, 0, 0, 0, 0, 0, 0, .1066, .2585, .3285, .2898

Finally, we are led to rediscover the corresponding generating functions. Usually, this is done via recurrence computations, and what we obtain here is equivalent to the EGF of [Stirling second kind \(partition\) numbers](#):

```
> for i to 5 do gfsolve(op(2,subs(r=i,surj)),labelled,z) od;
```

$$\{U(z) = e^z - 1, Z(z) = z, S(z) = e^z - 1\}$$

$$\{U(z) = e^z - 1, Z(z) = z, S(z) = (e^z)^2 - 2e^z + 1\}$$

$$\{U(z) = e^z - 1, Z(z) = z, S(z) = (e^z)^3 - 3(e^z)^2 + 3e^z - 1\}$$

$$\{U(z) = e^z - 1, Z(z) = z, S(z) = (e^z)^4 - 4(e^z)^3 + 6(e^z)^2 - 4e^z + 1\}$$

$$\{U(z) = e^z - 1, S(z) = (e^z)^5 - 5(e^z)^4 + 10(e^z)^3 - 10(e^z)^2 + 5e^z - 1, Z(z) = z\}$$

This suggests a pattern involving Pascal's triangle:

```
> Surj_z:=proc(k) local j;
```

```
  add(binomial(k,j)*(-1)^(k-j)*exp(z)^j,j=0..k) end;
```

```
> Surj_z(6); gfsolve(op(2,subs(r=6,surj)),labelled,z);
```

$$1 - 6e^z + 15(e^z)^2 - 20(e^z)^3 + 15(e^z)^4 - 6(e^z)^5 + (e^z)^6$$

$$\{U(z) = e^z - 1, Z(z) = z,$$

$$S(z) = 1 - 6e^z + 15(e^z)^2 - 20(e^z)^3 + 15(e^z)^4 - 6(e^z)^5 + (e^z)^6\}$$

For coefficients finally, we have by straight expansion

$$\sum_{j=0}^k \text{binomial}(k,j) (-1)^{(k-j)} j^n.$$

```
> Surj_nk:=proc(n,k) local j;
```

```
  'if'(n<>0,add(binomial(k,j)*(-1)^(k-j)*j^n,j=0..k),1) end;
```

The formula matches the values obtained by combstruct[gfseries]

```
> Order:=16: gfseries(op(2..3,subs(r=7,surj)),z);
```

```
table([
```

$$S(z) = z^7 + \frac{7}{2}z^8 + \frac{77}{12}z^9 + \frac{49}{6}z^{10} + \frac{1939}{240}z^{11} + \frac{4753}{720}z^{12} + \frac{1249}{270}z^{13} + \frac{77}{27}z^{14} + \frac{136111}{86400}z^{15}$$

+ O(z¹⁶)

$$U(z) = z + \frac{1}{2}z^2 + \frac{1}{6}z^3 + \frac{1}{24}z^4 + \frac{1}{120}z^5 + \frac{1}{720}z^6 + \frac{1}{5040}z^7 + \frac{1}{40320}z^8 + \frac{1}{362880}z^9 +$$

$$\frac{1}{3628800} z^{10} + \frac{1}{39916800} z^{11} + \frac{1}{479001600} z^{12} + \frac{1}{6227020800} z^{13} + \frac{1}{87178291200} z^{14} + \frac{1}{1307674368000} z^{15} + O(z^{16})$$

$$Z(z) = z$$

)

```
> seq(Surj_nk(n, 7) / n!, n=0..15);
```

```
1, 0, 0, 0, 0, 0, 0, 1,  $\frac{7}{2}$ ,  $\frac{77}{12}$ ,  $\frac{49}{6}$ ,  $\frac{1939}{240}$ ,  $\frac{4753}{720}$ ,  $\frac{1249}{270}$ ,  $\frac{77}{27}$ ,  $\frac{136111}{86400}$ 
```

Number of parts in integer compositions

This is the model IBDU.

```
> comp_r := [S, {S=Sequence(U, card=r), U=Sequence(Z, card>=1)}, unlabelled];
```

```
> for i to 5 do seq(count(subs(r=i, comp_r), size=m), m=0..15) od;
```

```
0, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1
```

```
0, 0, 1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14
```

```
0, 0, 0, 1, 3, 6, 10, 15, 21, 28, 36, 45, 55, 66, 78, 91
```

```
0, 0, 0, 0, 1, 4, 10, 20, 35, 56, 84, 120, 165, 220, 286, 364
```

```
0, 0, 0, 0, 0, 1, 5, 15, 35, 70, 126, 210, 330, 495, 715, 1001
```

```
> for i to 5 do subs(gfsolve(op(2, subs(r=i, comp_r)), labelled, z), S(z)) od;
```

$$-\frac{z}{-1+z} - \frac{z^2}{(-1+z)^2} - \frac{z^3}{(-1+z)^3} - \frac{z^4}{(-1+z)^4} - \frac{z^5}{(-1+z)^5}$$

The pattern is obvious and this corresponds to an explicit (and well-known!) binomial formula for the coefficients.

Number of parts in integer partitions

This is the model IBIU.

```
> part_r := [S, {S=Set(U, card=r), U=Sequence(Z, card>=1)}, unlabelled];
```

We can build tables, like before

```
> for i to 5 do seq(count(subs(r=i,part_r),size=m),m=0..15)
od;
```

```
0, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1
0, 0, 1, 1, 2, 2, 3, 3, 4, 4, 5, 5, 6, 6, 7, 7
0, 0, 0, 1, 1, 2, 3, 4, 5, 7, 8, 10, 12, 14, 16, 19
0, 0, 0, 0, 1, 1, 2, 3, 5, 6, 9, 11, 15, 18, 23, 27
0, 0, 0, 0, 0, 1, 1, 2, 3, 5, 7, 10, 13, 18, 23, 30
```

The generating functions are found by `combstruct[gfsolve]` to be rational:

```
> for i to 5 do gfsolve(op(2,subs(r=i,part_r)),unlabelled,z)
od;
```

$$\{Z(z) = z, U(z) = -\frac{z}{-1+z}, S(z) = -\frac{z}{-1+z}\}$$

$$\{Z(z) = z, U(z) = -\frac{z}{-1+z}, S(z) = \frac{z^2}{z^3 - z^2 - z + 1}\}$$

$$\{Z(z) = z, U(z) = -\frac{z}{-1+z}, S(z) = -\frac{z^3}{z^6 - z^5 - z^4 + z^2 + z - 1}\}$$

$$\{Z(z) = z, U(z) = -\frac{z}{-1+z}, S(z) = \frac{z^4}{z^{10} - z^9 - z^8 + 2z^5 - z^2 - z + 1}\}$$

$$\{Z(z) = z, U(z) = -\frac{z}{-1+z},$$

$$S(z) = -\frac{z^5}{z^{15} - z^{14} - z^{13} + z^{10} + z^9 + z^8 - z^7 - z^6 - z^5 + z^2 + z - 1}\}$$

The form is then easily inferred from the factored representations, as one recognizes cyclotomic polynomials.

```
> S5_z := subs(", S(z));
```

$$S5_z := -\frac{z^5}{z^{15} - z^{14} - z^{13} + z^{10} + z^9 + z^8 - z^7 - z^6 - z^5 + z^2 + z - 1}$$

```
> factor(S5_z); series(", z=0, 20);
```

$$-\frac{z^5}{(-1+z)^5 (z+1)^2 (z^2+z+1) (z^2+1) (z^4+z^3+z^2+z+1)}$$

$$z^5 + z^6 + 2z^7 + 3z^8 + 5z^9 + 7z^{10} + 10z^{11} + 13z^{12} + 18z^{13} + 23z^{14} + 30z^{15} + 37z^{16} + 47z^{17} + 57z^{18} + 70z^{19} + O(z^{20})$$

```
> z^5/mul(1-z^i,i=1..5); factor(""); series(", z=0, 20);
```

$$\frac{z^5}{(1-z)(1-z^2)(1-z^3)(1-z^4)(1-z^5)}$$

$$-\frac{z^5}{(-1+z)^5 (z+1)^2 (z^2+z+1) (z^2+1) (z^4+z^3+z^2+z+1)}$$

$$z^5 + z^6 + 2z^7 + 3z^8 + 5z^9 + 7z^{10} + 10z^{11} + 13z^{12} + 18z^{13} + 23z^{14} + 30z^{15} + 37z^{16} + 47z^{17} + 57z^{18} + 70z^{19} + O(z^{20})$$

■ Bounded capacity in the DBDU model (hashing)

We now consider configurations where the number of urns is a fixed number r , so that one can relax the constraint that urns need to be nonempty. We are thus dealing with the collection of the r^n functions of $[1 .. r]$ to $[1 .. r]$, not necessarily surjections. Such specifications also describe words (size being length) on an alphabet of cardinality r and this may be used to model hashing sequences in a table of length r . The set of all possible sequences (r fixed) is specified by

```
> hash_r := [S, {S=Prod(U$r), U=Set(Z)}, labelled] :
subs(r=10, hash_r) ;
[S, {S = Prod(U $ 10), U = Set(Z)}, labelled]
```

Here, we use a hack, with `Prod` instead of "Sequence" with fixed cardinality, since the current version of Combstruct does not accept Sequence applied to an argument that leads to an Epsilon structure, even in the case of a bounded cardinality modifier.

[We change the reduction procedure to take this new construct into account.

```
> lreduce := proc (e) eval (subs ({Set=proc () {args} end,
Prod=proc () [args] end}, e)) end:
ureduce := proc (e) eval (subs ({Set=proc () {[args]} end,
Sequence=proc () [args] end, Prod=proc () [args] end}, e))
end:
```

The number of objects of size n is, as predicted, 10^n .

```
> seq(count(subs(r=10, hash_r), size=j), j=0..10) ;
1, 10, 100, 1000, 10000, 100000, 1000000, 10000000, 100000000, 1000000000,
10000000000
```

Next, we consider urns with a bounded maximum capacity b :

```
> hash_br := [S, {S=Prod(U$r), U=Set(Z, card<=b)}, labelled] :
subs({r=5, b=3}, hash_br) ;
[S, {S = Prod(U $ 5), U = Set(Z, card ≤ 3)}, labelled]
```

Such specifications also describe words (size being length) on an alphabet of cardinality r , given that no letter is used more than b times. This models hashing sequences in a table of length r when pages have capacity b .

```
> lreduce(draw(subs(r=10, hash_r), size=30)) ;
[{Z25, Z28}, {Z9, Z17, Z5, Z30, Z23}, {Z29, Z6, Z13}, {Z7}, {Z14, Z1, Z18, Z24, Z22},
{Z21, Z10, Z11}, E, {Z19, Z2, Z16, Z26, Z27}, {Z4, Z20, Z12}, {Z3, Z15, Z8}]
> lreduce(draw(subs({r=10, b=4}, hash_br), size=30)) ;
[{Z14, Z9, Z24}, {Z19, Z5, Z22, Z15}, {Z2, Z6, Z16, Z27}, {Z23, Z26}, {Z29, Z25},
{Z21, Z7, Z30, Z18}, {Z1, Z4, Z17}, {Z3, Z12}, {Z10, Z11, Z20}, {Z28, Z13, Z8}]
```

The following command automatically creates a table of maximum urn occupancy: the j -th entry in the b -th line is the probability that j balls thrown into 10 urns fit into urns of

capacity b .

```
> for i 1 to 6 do
  seq(evalf(count(subs({r=10,b=i},hash_br),size=j)/count(sub
s(r=10,hash_r),size=j),4),j=0..20); od;
Syntax error, unexpected number
```

Naturally, Combstruct automatically recognizes "impossible" configurations:

```
> draw(subs({r=10,b=3},hash_br),size=31);
Error, (in combstruct/drawgrammar) there is no structure of this size
```

■ Bounded capacity in the IBDU model (bosons)

This is a model of integer compositions. (A more sophisticated example that is related to submarine detection is treated below.) We consider here a fixed number r of urns and this is exactly the so-called "Bose-Einstein" model of statistical physics, where the corresponding objects are called bosons.

Mark Kobrak, a chemist at the University of Chicago wrote to us (December 13, 1996):

The basic problem is this: I am interested in simulating a problem in laser spectroscopy. A molecule has n modes, and I need to generate every possible combination which places up to m quanta in each mode. As I go through each one, I will analyze its energy.

This is exactly like a problem where, given n jars, you may put up to m marbles in each jar. The marbles are indistinguishable. I know how to count the number of combinations, but I need a good way to program a computer to go through all these combinations.

We let again r denote the number of urns and b the bucket capacity, i.e., the maximum number of urns that may fit into any given urn. We have the model of integer compositions:

```
> boson_br := [S, {S=Prod(U$r), U=Sequence(Z, card<=b)},
unlabelled]: subs({r=5,b=3},boson_br);
[S, {U=Sequence(Z, card ≤ 3), S=Prod(U $ 5)}, unlabelled]
```

For instance, here are all the 95 possible ways of putting j marbles ($j \leq 4$) into $r = 5$ jars, each jar being of maximum capacity $b = 2$.

```
> seq(count(subs({r=5,b=2},boson_br),size=j),j=1..4);
5, 15, 30, 45
> for j to 4 do
  j=map(ureduce,allstructs(subs({r=5,b=2},boson_br),size=j))
  od;
1 = [[[Z], E, E, E, E], [E, E, E, [Z], E], [E, E, [Z], E, E], [E, E, E, E, [Z]],
[E, [Z], E, E, E]]
2 = [[[Z], E, E, [Z], E], [E, E, [Z, Z], E, E], [E, [Z], [Z], E, E], [E, E, E, [Z, Z], E],
[E, E, E, E, [Z, Z]], [E, [Z], E, E, [Z]], [[Z], E, [Z], E, E], [E, E, E, [Z], [Z]],
[E, [Z, Z], E, E, E], [[Z, Z], E, E, E, E], [E, [Z], E, [Z], E], [[Z], [Z], E, E, E],
[E, E, [Z], E, [Z]], [E, E, [Z], [Z], E], [[Z], E, E, E, [Z]]]
3 = [[[Z], [Z, Z], E, E, E], [E, E, [Z], [Z], [Z]], [[Z], E, E, [Z], [Z]],
[[Z], E, [Z], E, [Z]], [[Z], [Z], E, [Z], E], [E, [Z, Z], E, E, [Z]]]
```

```

[E, [Z, Z], E, [Z], E], [[Z, Z], E, [Z], E, E], [E, [Z], [Z], [Z], E],
[E, E, [Z], [Z, Z], E], [[Z], E, [Z, Z], E, E], [E, [Z], [Z, Z], E, E],
[E, E, [Z, Z], E, [Z]], [[Z, Z], [Z], E, E, E], [[Z, Z], E, E, E, [Z]],
[E, [Z], E, E, [Z, Z]], [[Z], E, E, [Z, Z], E], [[Z], E, E, E, [Z, Z]],
[E, [Z, Z], [Z], E, E], [E, [Z], E, [Z, Z], E], [[Z], [Z], [Z], E, E],
[E, E, [Z], E, [Z, Z]], [[Z, Z], E, E, [Z], E], [E, [Z], [Z], E, [Z]],
[[Z], [Z], E, E, [Z]], [E, E, E, [Z, Z], [Z]], [E, [Z], E, [Z], [Z]],
[[Z], E, [Z], [Z], E], [E, E, E, [Z], [Z, Z]], [E, E, [Z, Z], [Z], E]]
4 = [[[Z, Z], E, [Z], E, [Z]], [E, [Z, Z], E, E, [Z, Z]], [[Z], [Z], [Z, Z], E, E],
[E, E, [Z], [Z], [Z, Z]], [[Z], [Z, Z], [Z], E, E], [[Z], E, [Z], [Z, Z], E],
[E, [Z, Z], [Z], [Z], E], [E, E, [Z, Z], [Z, Z], E], [E, [Z, Z], E, [Z], [Z]],
[[Z, Z], E, [Z], [Z], E], [[Z], [Z], E, E, [Z, Z]], [E, [Z, Z], [Z], E, [Z]],
[[Z, Z], [Z], E, E, [Z]], [E, E, E, [Z, Z], [Z, Z]], [E, E, [Z, Z], [Z], [Z]],
[E, [Z], [Z, Z], [Z], E], [[Z], E, [Z], [Z], [Z]], [[Z], E, [Z, Z], E, [Z]],
[E, [Z], [Z], E, [Z, Z]], [E, [Z, Z], E, [Z, Z], E], [[Z], [Z, Z], E, [Z], E],
[E, E, [Z], [Z, Z], [Z]], [[Z, Z], E, E, [Z, Z], E], [[Z], [Z], [Z], [Z], E],
[[Z], E, E, [Z, Z], [Z]], [E, [Z], E, [Z], [Z, Z]], [[Z], [Z], E, [Z, Z], E],
[[Z, Z], E, [Z, Z], E, E], [E, [Z], [Z], [Z, Z], E], [[Z, Z], [Z], [Z], E, E],
[[Z], [Z], [Z], E, [Z]], [[Z], [Z, Z], E, E, [Z]], [[Z, Z], [Z, Z], E, E, E],
[[Z], E, [Z, Z], [Z], E], [[Z], [Z], E, [Z], [Z]], [E, E, [Z, Z], E, [Z, Z]],
[[Z], E, [Z], E, [Z, Z]], [E, [Z], [Z, Z], E, [Z]], [E, [Z, Z], [Z, Z], E, E],
[E, [Z], E, [Z, Z], [Z]], [[Z, Z], E, E, [Z], [Z]], [E, [Z], [Z], [Z], [Z]],
[[Z, Z], [Z], E, [Z], E], [[Z, Z], E, E, E, [Z, Z]], [[Z], E, E, [Z], [Z, Z]]]

```

Stack polyominoes

Polyominoes are familiar objects of combinatorial mathematics. A stack polyomino or *stack* is a piling up of nonempty integer intervals, each interval being included in the previous one. The number of intervals comprising a stack are called the height; the total length of the intervals is called the size. For instance, the collection

```

[ > [1, 12], [3, 8], [4, 7], [4, 7], [6, 7];
      [1, 12], [3, 8], [4, 7], [4, 7], [6, 7]

```

is a stack of height 5 and size

```

[ > (12-1) + (8-3) + (7-4) + (7-4) + (7-6);
      23

```

We may assume stacks to be left justified, starting at 1. Stacks of height exactly r are specified by ($r = 5$ in the example)

```

[ > stack := {st, {st=Prod(left, right), left=Set(U, card=r), right=Set(
      U, card<r), U=Sequence(Z, card>=1)}, unlabelled};
      subs(r=5, stack);

```

$[st, \{ \text{left} = \text{Set}(U, \text{card} = 5), \text{right} = \text{Set}(U, \text{card} < 5), st = \text{Prod}(\text{left}, \text{right}),$
 $U = \text{Sequence}(\mathbb{Z}, 1 \leq \text{card}) \}, \text{unlabelled}]$

Combinatorially, we view a stack as an ascending staircase (the left part) of height r followed by a descending staircase (the right part) of height at most r . Staircases of fixed height are defined as partitions into bounded summands.

The distribution of height in stacks is for small height:

```
> for i to 6 do seq(count(subs(r=i,stack),size=m),m=0..20) od;
      0, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1
      0, 0, 1, 2, 4, 6, 9, 12, 16, 20, 25, 30, 36, 42, 49, 56, 64, 72, 81, 90, 100
      0, 0, 0, 1, 2, 5, 9, 16, 25, 39, 56, 80, 109, 147, 192, 249, 315, 396, 489, 600, 726
      0, 0, 0, 0, 1, 2, 5, 10, 19, 32, 54, 84, 129, 190, 275, 386, 536, 726, 973, 1282, 1672
      0, 0, 0, 0, 0, 1, 2, 5, 10, 20, 35, 61, 99, 159, 244, 369, 541, 784, 1109, 1551, 2131
      0, 0, 0, 0, 0, 0, 1, 2, 5, 10, 20, 36, 64, 106, 174, 274, 425, 640, 952, 1384, 1989
```

The total number of stacks is:

```
> for i to 6 do for m from 0 to 10 do
      stnum[i,m]:=count(subs(r=i,stack),size=m) od od: for m to 6
      do m,convert([seq(stnum[i,m],i=1..6)],'+') od;
      1, 1
      2, 2
      3, 4
      4, 8
      5, 15
      6, 27
```

This is **M1102** of the *Encyclopedia of Integer Sequences*. The table given there is incomplete.

However, from an earlier computation of generating functions, we have a formula for the generating function of all stacks:

```
> Q:=Product(1-z^k,k=1..n);
      Stack_z:=1+Sum(z^r/subs(n=r,Q)/subs(n=r-1,Q),r=1..infinity);
```

$$Q := \prod_{k=1}^n (1 - z^k)$$

$$\text{Stack}_z := 1 + \left(\sum_{r=1}^{\infty} \frac{z^r}{\left(\prod_{k=1}^r (1 - z^k) \right) \left(\prod_{k=1}^{r-1} (1 - z^k) \right)} \right)$$

```
> Order:=30: series(value(subs(infinity=Order+3,Stack_z)),z=0);
1 + z + 2 z^2 + 4 z^3 + 8 z^4 + 15 z^5 + 27 z^6 + 47 z^7 + 79 z^8 + 130 z^9 + 209 z^10 + 330 z^11 + 512
z^12 + 784 z^13 + 1183 z^14 + 1765 z^15 + 2604 z^16 + 3804 z^17 + 5504 z^18 + 7898 z^19 + 11240 z^20
+ 15880 z^21 + 22277 z^22 + 31048 z^23 + 43003 z^24 + 59220 z^25 + 81098 z^26 + 110484 z^27 +
149769 z^28 + 202070 z^29 + O(z^30)
```


■ A problem in submarine detection

Problem 68-16 that appeared in the 1968 volume of *SIAM Review* reads as follows:

Problem 68-16. A combinatorial Problem, by Melda Hayes (Ocean Technology Inc.).

In how many ways can n identical balls be distributed in r boxes in a row such that each pair of adjacent boxes contains at least 4 balls? This problem arose in some work on submarine detection.

The problem is thus relative to integer compositions with constrained summands. The constraints are reminiscent of maximum capacity problems but they concern successive summands. For pedagogical reasons, we decompose the solution in two phases:

- a problem of counting words over a 5-letter alphabet that translates the succession constraint;
- the original problem.

■ The word counting problem

Consider an alphabet comprising letters

```
> A=seq(a..j, j=0..4);  
A = (a0, a1, a2, a3, a4)
```

There a_j , for $j < 4$ represents symbolically a summand of size j ; the quantity a_4 represents a summand of cardinality *at least* 4. The first grammar specifies words over the alphabet A such that the sum of indices of any two consecutive letters is at least 4.

```
> grammar1 := [Q4, {  
  seq(Q.i=Union(Epsilon, seq(Prod(a..j, Q..j), j=4-i..4)), i=0..4)  
,  
  seq(a..j=Z, j=0..4)  
}, unlabelled];  
grammar1 := [Q4, { Q0 = Union(E, Prod(a4, Q4)),  
  Q1 = Union(E, Prod(a3, Q3), Prod(a4, Q4)),  
  Q2 = Union(E, Prod(a2, Q2), Prod(a3, Q3), Prod(a4, Q4)), a1 = Z, a3 = Z, a2 = Z,  
  a4 = Z, a0 = Z, Q4 =  
  Union(E, Prod(a0, Q0), Prod(a1, Q1), Prod(a2, Q2), Prod(a3, Q3), Prod(a4, Q4)),  
  Q3 = Union(E, Prod(a1, Q1), Prod(a2, Q2), Prod(a3, Q3), Prod(a4, Q4)) },  
  unlabelled]
```

The grammar above is just a translation of the finite automaton that recognizes the language of symbolic constraints. We can solve the counting problem easily.

```
> seq(count(grammar1, size=j), j=0..20);  
1, 5, 15, 55, 190, 671, 2353, 8272, 29056, 102091, 358671, 1260143, 4427294,  
15554592, 54648506, 191998646, 674555937, 2369942427, 8326406594,  
29253473175, 102777312308  
> gfsolve(op(2..3, grammar1), z);
```

$$\{ Z(z) = z, Q4(z) = -\frac{-z^3 + 1 + 2z + z^4 - 3z^2}{-1 + 3z^2 - z^4 - 4z^3 + 3z + z^5}, a3(z) = z, a4(z) = z,$$

$$Q3(z) = \frac{-1 - z + z^2}{-1 + 3z^2 - z^4 - 4z^3 + 3z + z^5}, a2(z) = z,$$

$$Q2(z) = -\frac{1}{-1 + 3z^2 - z^4 - 4z^3 + 3z + z^5}, a0(z) = z,$$

$$Q0(z) = -\frac{1 - z^2 + z^3 - 2z}{-1 + 3z^2 - z^4 - 4z^3 + 3z + z^5}, a1(z) = z,$$

$$Q1(z) = \frac{-1 + z}{-1 + 3z^2 - z^4 - 4z^3 + 3z + z^5} \}$$

The generating function of all words has been found to be

```
> Q_z := factor(subs(", Q4(z)));
```

$$Q_z := -\frac{(-1 + z)(z^3 - 3z - 1)}{-1 + 3z^2 - z^4 - 4z^3 + 3z + z^5}$$

```
> series(Q_z, z=0, 20);
```

$$1 + 5z + 15z^2 + 55z^3 + 190z^4 + 671z^5 + 2353z^6 + 8272z^7 + 29056z^8 + 102091z^9 + 358671z^{10} + 1260143z^{11} + 4427294z^{12} + 15554592z^{13} + 54648506z^{14} + 191998646z^{15} + 674555937z^{16} + 2369942427z^{17} + 8326406594z^{18} + 29253473175z^{19} + O(z^{20})$$

The asymptotics results from locating the dominant pole:

```
> Digits:=30: fsolve(denom(Q_z), z);
```

```
-1.68250706566236233772362329784, -.830830026003772851058548298459,
.284629676546570280887585337233, 1.30972146789057012811385014493,
1.91898594722899477978073611413
```

```
> rho:=fsolve(denom(Q_z), z, 0..1);
```

```
rho := .284629676546570280887585337233
```

```
> c:=-subs(z=rho, numer(Q_z)/diff(denom(Q_z), z))/rho;
```

```
c := 1.25170169910163367947646373163
```

```
> Q_n_asympt:=proc(n) round(c*rho^(-n)) end;
```

```
> count(grammar1, size=30); Q_n_asympt(30); evalf("/");
```

```
29450689289430149
```

```
29450689289430243
```

```
1.000000000000000319177588939273
```

The integer composition problem

For the original problem, we only need to interpret $a.j$ for $j < 4$ as meaning a summand of value j , that is to say Z repeated j times, and $a4$ as a summand of value at least 4.

```
> grammar2 := [Q4, {
```

```
seq(Q.i=Union(Epsilon, seq(Prod(a.j, Q.j), j=4-i..4)), i=0..4)
```

```

seq(a.j=Sequence(Z, card=j), j=0..3), a4=Sequence(Z, card>=4)
}, unlabelled];
grammar2 := [Q4, {Q0 = Union(E, Prod(a4, Q4)), a4 = Sequence(Z, 4 ≤ card),
a0 = Sequence(Z, card = 0), a3 = Sequence(Z, card = 3), a2 = Sequence(Z, card = 2),
a1 = Sequence(Z, card = 1), Q1 = Union(E, Prod(a3, Q3), Prod(a4, Q4)),
Q2 = Union(E, Prod(a2, Q2), Prod(a3, Q3), Prod(a4, Q4)), Q4 =
Union(E, Prod(a0, Q0), Prod(a1, Q1), Prod(a2, Q2), Prod(a3, Q3), Prod(a4, Q4)),
Q3 = Union(E, Prod(a1, Q1), Prod(a2, Q2), Prod(a3, Q3), Prod(a4, Q4))},
unlabelled]
> seq(count(grammar2, size=j), j=0..30);
2, 1, 1, 1, 7, 11, 17, 25, 51, 94, 165, 280, 496, 887, 1576, 2770, 4880, 8630, 15276,
26990, 47656, 84183, 148781, 262921, 464528, 820699, 1450091, 2562250, 4527272,
7999104, 14133456

```

We then get the generating function as before.

```

> gr2_sys:=gfsolve(op(2..3, grammar2), z);
gr2_sys := {Z(z) = z, a0(z) = 1, a4(z) = - $\frac{z^4}{-1+z}$ ,
Q4(z) =  $\frac{-z^4 - z^2 + 2 + z^6 - 2z^3 + z}{z^9 + z^8 - z^6 - 3z^5 - 3z^4 - z^3 - z^2 + 1}$ , a3(z) = z^3,
Q3(z) = - $\frac{z^3 - z - 1}{z^9 + z^8 - z^6 - 3z^5 - 3z^4 - z^3 - z^2 + 1}$ , a2(z) = z^2,
Q2(z) =  $\frac{1}{z^9 + z^8 - z^6 - 3z^5 - 3z^4 - z^3 - z^2 + 1}$ ,
Q0(z) =  $\frac{z^6 - z^4 - z^3 - z^2 + 1}{z^9 + z^8 - z^6 - 3z^5 - 3z^4 - z^3 - z^2 + 1}$ , Q1(z) = - $\frac{-1+z}{z^8 - z^5 - 2z^4 - z^3 - z + 1}$ ,
a1(z) = z}
> QQ_z:=factor(subs(", Q4(z)));
QQ_z :=  $\frac{(-1+z)(z^5 + z^4 - 2z^2 - 3z - 2)}{(z+1)(z^8 - z^5 - 2z^4 - z^3 - z + 1)}$ 
> series(QQ_z, z=0, 31);
2 + z + z^2 + z^3 + 7z^4 + 11z^5 + 17z^6 + 25z^7 + 51z^8 + 94z^9 + 165z^10 + 280z^11 + 496z^12
+ 887z^13 + 1576z^14 + 2770z^15 + 4880z^16 + 8630z^17 + 15276z^18 + 26990z^19 + 47656
z^20 + 84183z^21 + 148781z^22 + 262921z^23 + 464528z^24 + 820699z^25 + 1450091z^26 +
2562250z^27 + 4527272z^28 + 7999104z^29 + 14133456z^30 + O(z^31)
> rho:=fsolve(denom(QQ_z), z, 0..1);
rho := .565964944350751354426781593907

```

```

> QQ_n_asympt:=proc(n)
  round(subs(z=rho, -1/diff(denom(QQ_z), z)/rho*numer(QQ_z))*r
  ho^(-n)) end: QQ_n_asympt(n);
round(
  .541922081536031689337273382179 .565964944350751354426781593907(-n))
> count(grammar2, size=40); QQ_n_asympt(40); evalf("/");
4191365486
4191365824
.999999919358029293317060744350

```

This solves the original problem. A detailed solution involving reduction of infinite matrices was submitted by D. R. Breach, University of Toronto.

Fixing the number of boxes

The published solution by D. R. Breach also provided detailed tables for number of balls (size) $n \leq 20$ and number of boxes $r \leq 10$. Like before, we could generate specifications for each given value of r . However, it is possible to solve simultaneously *all* such problems by means of bivariate generating functions. Roughly, Combstruct makes it possible to insert marks (in the form of Epsilons that do not modify the counting results).

The specification *grammar1* generates objects where all atoms are anonymously labelled Z.

```

> allstructs(grammar1, size=2);
[Prod(Z, Prod(Z, E)), Prod(Z, Prod(Z, E)), Prod(Z, Prod(Z, E)),
  Prod(Z, Prod(Z, E)), Prod(Z, Prod(Z, E)), Prod(Z, Prod(Z, E)), Prod(Z, Prod(Z, E)),
  Prod(Z, Prod(Z, E)), Prod(Z, Prod(Z, E)), Prod(Z, Prod(Z, E)), Prod(Z, Prod(Z, E)),
  Prod(Z, Prod(Z, E)), Prod(Z, Prod(Z, E)), Prod(Z, Prod(Z, E)), Prod(Z, Prod(Z, E))
]

```

If we wish to keep track of additional informations, we can just take products with suitable structures of size 0 (Epsilons). This is a general programming technique of Combstruct. Here we use *b.j* to mark an occurrence of an *a.j*.

```

> grammar1bis := [Q4, {
  seq(Q.i=Union(Epsilon, seq(Prod(a.j, Q.j), j=4-i..4)), i=0..4)
  ,
  seq(a.j=Prod(Z, b.j), j=0..4),
  seq(b.j=Epsilon, j=0..4)
}, unlabelled];
grammar1bis := [Q4, { Q0 = Union(E, Prod(a4, Q4)),
  Q1 = Union(E, Prod(a3, Q3), Prod(a4, Q4)),
  Q2 = Union(E, Prod(a2, Q2), Prod(a3, Q3), Prod(a4, Q4)), Q4 =
  Union(E, Prod(a0, Q0), Prod(a1, Q1), Prod(a2, Q2), Prod(a3, Q3), Prod(a4, Q4)),
  Q3 = Union(E, Prod(a1, Q1), Prod(a2, Q2), Prod(a3, Q3), Prod(a4, Q4)), b1 = E,
  b2 = E, b3 = E, b4 = E, b0 = E, a0 = Prod(Z, b0), a1 = Prod(Z, b1), a2 = Prod(Z, b2),
  a3 = Prod(Z, b3), a4 = Prod(Z, b4) }, unlabelled]

```

```

> subs({Prod='', Epsilon=NULL}, allstructs(grammar1bis, size=2)
);
[((Z, b3), ((Z, b3))), ((Z, b2), ((Z, b3))), ((Z, b4), ((Z, b2))),
  ((Z, b4), ((Z, b4))), ((Z, b2), ((Z, b4))), ((Z, b4), ((Z, b1))), ((Z, b4), ((Z, b0))),
  ((Z, b3), ((Z, b1))), ((Z, b3), ((Z, b4))), ((Z, b1), ((Z, b4))), ((Z, b1), ((Z, b3))),
  ((Z, b2), ((Z, b2))), ((Z, b0), ((Z, b4))), ((Z, b3), ((Z, b2))), ((Z, b4), ((Z, b3)))
]

```

Likewise, we may insert a generic marker b for each summand in the compositions described by *grammar2*:

```

> grammar2bis := [Q4, {
  seq(Q.i=Union(Epsilon, seq(Prod(a.j, Q.j), j=4-i..4)), i=0..4)
  ,
  seq(a.j=Prod(b, Sequence(Z, card=j)), j=0..3), a4=Prod(b, Sequence(Z, card>=4)),
  b=Epsilon
}, unlabelled];
grammar2bis := [Q4, {Q0 = Union(E, Prod(a4, Q4)),
  a0 = Prod(b, Sequence(Z, card = 0)), a4 = Prod(b, Sequence(Z, 4 ≤ card)),
  Q1 = Union(E, Prod(a3, Q3), Prod(a4, Q4)),
  Q2 = Union(E, Prod(a2, Q2), Prod(a3, Q3), Prod(a4, Q4)), b = E, Q4 =
  Union(E, Prod(a0, Q0), Prod(a1, Q1), Prod(a2, Q2), Prod(a3, Q3), Prod(a4, Q4)),
  Q3 = Union(E, Prod(a1, Q1), Prod(a2, Q2), Prod(a3, Q3), Prod(a4, Q4)),
  a2 = Prod(b, Sequence(Z, card = 2)), a1 = Prod(b, Sequence(Z, card = 1)),
  a3 = Prod(b, Sequence(Z, card = 3))}, unlabelled]

```

The functions `combstruct[gfeqns]` and `combstruct[gfsolve]` respect such marks and allow for the possibility of assigning auxiliary variables for such marks. Thus, one can determine multivariate generating functions. Here u is the variable associated to mark b .

```

> gfeqns(op(2..3, grammar2bis), z, [[u, b]]);
[
Q4(z, u) = 1 + a0(z, u) Q0(z, u) + a1(z, u) Q1(z, u) + a2(z, u) Q2(z, u)
  + a3(z, u) Q3(z, u) + a4(z, u) Q4(z, u), Q0(z, u) = 1 + a4(z, u) Q4(z, u), Q3(z, u) =
1 + a1(z, u) Q1(z, u) + a2(z, u) Q2(z, u) + a3(z, u) Q3(z, u) + a4(z, u) Q4(z, u),
Q1(z, u) = 1 + a3(z, u) Q3(z, u) + a4(z, u) Q4(z, u), a1(z, u) = b(z, u) Z(z, u),
b(z, u) = u, a2(z, u) = b(z, u) Z(z, u)2,
a4(z, u) = b(z, u)  $\left( \frac{1}{1 - Z(z, u)} - 1 - Z(z, u) - Z(z, u)^2 - Z(z, u)^3 \right)$ 
Q2(z, u) = 1 + a2(z, u) Q2(z, u) + a3(z, u) Q3(z, u) + a4(z, u) Q4(z, u), Z(z, u) = z,

```

$$a_3(z, u) = b(z, u) Z(z, u)^3, a_0(z, u) = b(z, u)$$

Solving for $Q_4(z)$, we find:

$$QQ_zu := (-u^2 z^2 - 2u^2 z^3 + u^4 z^6 + u + 1 - u^3 z^4 + u z) (-1 + z) / (u z^2 - 1) / (u^4 z^8 - u^2 z^5 - 2u^2 z^4 - u z^3 - z + 1);$$

$$QQ_zu := \frac{(-u^2 z^2 - 2u^2 z^3 + u^4 z^6 + u + 1 - u^3 z^4 + u z) (-1 + z)}{(u z^2 - 1) (u^4 z^8 - u^2 z^5 - 2u^2 z^4 - u z^3 - z + 1)}$$

We then automatically obtained the main table in the solution published by *SIAM Review*.

We can even detect errors: for instance, in the last line we must have 1261 (instead of 1211) and 4756 (instead of 4762)!

$$> \text{map}(\text{series}, \text{series}(QQ_zu, z=0, 21), u, \text{infinity});$$

$$(1 + u) + (u) z + (u) z^2 + (u) z^3 + (u + 5 u^2 + u^3) z^4 + (u + 6 u^2 + 4 u^3) z^5 + (u + 7 u^2 + 9 u^3) z^6 + (u + 8 u^2 + 16 u^3) z^7 + (u + 9 u^2 + 25 u^3 + 15 u^4 + u^5) z^8 + (u + 10 u^2 + 35 u^3 + 40 u^4 + 8 u^5) z^9 + (u + 11 u^2 + 46 u^3 + 76 u^4 + 31 u^5) z^{10} + (u + 12 u^2 + 58 u^3 + 124 u^4 + 85 u^5) z^{11} + (u + 13 u^2 + 71 u^3 + 185 u^4 + 190 u^5 + 35 u^6 + u^7) z^{12} + (u + 14 u^2 + 85 u^3 + 260 u^4 + 360 u^5 + 154 u^6 + 13 u^7) z^{13} + (u + 15 u^2 + 100 u^3 + 350 u^4 + 610 u^5 + 424 u^6 + 76 u^7) z^{14} + (u + 16 u^2 + 116 u^3 + 456 u^4 + 956 u^5 + 930 u^6 + 295 u^7) z^{15} + (u + 17 u^2 + 133 u^3 + 579 u^4 + 1415 u^5 + 1775 u^6 + 889 u^7 + 70 u^8 + u^9) z^{16} + (u + 18 u^2 + 151 u^3 + 720 u^4 + 2005 u^5 + 3080 u^6 + 2188 u^7 + 448 u^8 + 19 u^9) z^{17} + (u + 19 u^2 + 170 u^3 + 880 u^4 + 2745 u^5 + 4985 u^6 + 4652 u^7 + 1669 u^8 + 155 u^9) z^{18} + (u + 20 u^2 + 190 u^3 + 1060 u^4 + 3655 u^5 + 7650 u^6 + 8891 u^7 + 4718 u^8 + 805 u^9) z^{19} + (u + 21 u^2 + 211 u^3 + 1261 u^4 + 4756 u^5 + 11256 u^6 + 15686 u^7 + 11201 u^8 + 3136 u^9 + 126 u^{10} + u^{11}) z^{20} + (O(u^0)) z^{21}$$

We have also easy access to any subproblem defined by fixing the number r of boxes:

$$> \text{map}(\text{normal}, \text{series}(QQ_zu, u=0, 8));$$

$$1 - \frac{1}{-1 + z} u - \frac{z^4 (-5 + 4 z)}{(-1 + z)^2} u^2 + \frac{z^4 (-1 - z + z^5)}{(-1 + z)^3} u^3 + \frac{z^8 (15 - 20 z + 6 z^2)}{(-1 + z)^4} u^4 - \frac{z^8 (1 + 3 z + z^2 - 11 z^5 + 7 z^6)}{(-1 + z)^5} u^5 + \frac{z^{12} (35 - 56 z + 25 z^2 - 4 z^3 + z^6)}{(-1 + z)^6} u^6 + \frac{z^{12} (-1 - 6 z - 6 z^2 - z^3 + 66 z^5 - 74 z^6 + 21 z^7)}{(-1 + z)^7} u^7 + O(u^8)$$